Fortran-77 Computer Programs for 2D and 2.5D analysis of gravity anomalies with variable density contrast

D. BHASKARA RAO AND M. J. PRAKASH

Department of Geophysics, Andhra University, Visakhapatnam, India

(Received March 27, 1995; accepted February 10, 1997)

Abstract. Three computer programs in Fortran 77 are presented to calculate the gravity anomalies of 2.5-D and 2-D bodies with either constant density contrast or linear decrease of density contrast with depth or a second order decrease of density contrast with depth. The decrease of density contrast in sedimentary basins may be approximated by a quadratic function. A sedimentary basin may be viewed as a number of 2-D or 2.5-D prisms placed in juxtaposition. Approximate equations for both cases are developed for rapid calculation of gravity anomalies. Efficient methods are developed for two-and-half and two-dimensional analysis of gravity anomalies by an appropriate use of the exact and approximate equations. The depths to the basement are adjusted iteratively by comparing the calculated anomalies with the observed anomalies. A computer program is also developed to deduce the structure of a sedimentary basin along any off-set profile by a suitable modification of the anomaly equations. By interpreting several off-set profiles covering the entire basin, a depth contour map can be drawn which is comparable to a structural map deduced by 3-D methods. Two field examples are given demonstrating the utility of the computer programs.

1. Introduction

The interpretation of gravity anomalies may be carried out with the assumption that the causative body is two-dimensional. Many mehods are available in the literature to calculate gravity anomalies of two-dimensional bodies. Talwani et al. (1959) developed methods to calculate the gravity anomalies of 2-D bodies of arbitrary shape. Bott (1960) has developed a method for modelling of gravity anomalies over sedimentary basins. But in nature two-dimensional bodies rarely occur and most bodies have finite strike length. There are several algorithms available in the literature for interpretation of gravity anomalies of three-dimensional bodies. Nagy (1966), Talwani and Ewing (1960), Cordell and Henderson (1968) have developed methods for 3-D

Corresponding author: M. J. Prakash; Dpt. Geophysics, Andhra University, Visakhapatnam 530003 (A.P.), India



Fig. 1 - The linking of the subroutines with the main program.

modelling of gravity anomalies. Goetze and Lahmeyer (1988) have developed the 3-D interactive modelling for gravity and magnetic anomalies. Many causative bodies can be assumed to have a uniform vertical cross-section and terminate at a finite distance, and these may be called 2-D bodies of finite strike length or 2.5-D bodies. Though the 3-D programs can be used for modelling of gravity anomalies over 2.5-D bodies, they require clearly a lot of computer time. Instead, methods can be developed for modelling the gravity anomalies over such bodies along a profile taken perpendicular to its strike length by deriving suitable equations.

The equations for the gravity anomaly of 2.5-D polygon bodies with constant density contrast were determined by Rasmussen and Pedersen (1979) and a computer program for calculation of these anomalies was presented by Enmark (1981). Bhaskara Rao (1986) has shown that the decrease of density contrast in sedimentary basins can be approximated by a quadratic density function. Bhaskara Rao, Prakash and Ramesh Babu (1990) presented 3-D and 2.5-D modelling techniques of gravity anomalies with a quadratic density function. The approximation of the decrease of density contrast in sedimentary basins by a quadratic function makes it possible to derive a closed form solution in the space domain. The principle is used here to develop the generalized computer programs GR2HDSTR for 2.5-D, and GR2DSTR for 2-D inversion of gravity data either for a constant density contrast or for a variable density contrast with depth. A computer program GA3DSTR is also presented to deduce the structure of a sedimentary basin along any off-set profile, and hence to construct a contour map for basement depths.

2. Anomaly equations

The decrease of density contrast in many sedimentary basins could be fairly accurately approximated by a quadratic function (Bhaskara Rao, 1986) as

$$\Delta\rho(z) = a_0 + a_1 z + a_2 z^2, \tag{1}$$

where z represents the depth measured positive in the downward direction, a_0 represent the extrapolated value of density contrast at the surface, and a_1 and a_2 are the constants of the quadratic function.

For 2.5-D analysis of the gravity anomalies, a sedimentary basin may be viewed as a number of vertical prisms of half thickness T, and half strike length Y, symmetrically placed below each observation point along a profile. Thus, the equation for the gravity anomaly of a prismatic model of half strike length Y is given by Bhaskara Rao et al. (1990) as,

$$\begin{aligned}\Delta g(x) = 2\gamma a_0 & \left[\left[z \arctan\frac{YF}{zr} + \frac{F}{2}\, 1n\, \frac{r-Y}{r+Y} + \frac{Y}{2}\, 1n\, \frac{r-F}{r+F}\right]_{z=Z_1}^{Z_2}\right]_{u=-T}^{T} \\ + \frac{2\gamma a_1}{2} & \left[\left[z^2 \arctan\frac{YF}{zr} + 2YF\, 1n\, 2(r+z) - Y^2 \arctan\frac{Fz}{Yr} - F^2 \arctan\frac{Yz}{Fr}\right]_{z=Z_1}^{Z_2}\right]_{u=-T}^{T} \\ + \frac{2\gamma a_2}{3} & \left[\left[z^3 \arctan\frac{YF}{zr} + 2YFr - \frac{F^3}{2}\, 1n\, \frac{r-Y}{r+Y} - \frac{Y^3}{2}\, 1n\, \frac{r-F}{r+F}\right]_{z=Z_1}^{Z_2}\right]_{u=-T}^{T}\end{aligned} \tag{2}$$

where $F = u\text{-}x$, $r = \sqrt{(u\text{-}x)^2 + z^2 + Y^2}$, and γ is the gravitational constant. The other parameters are shown in Fig. 2. The corresponding equation for a 2-D prismatic model is given by Bhaskara Rao (1986).

3. Approximate equation

As computation of the gravity anomaly at any point involves repeated use of Eq. (2), it necessarily takes a lot of computer time. In order to save computer time and make the programs of practical use, we have developed an approximate equation for anomaly calculation. This is valid for a small distance from the centre of the prism, and is obtained by treating the prism as a sheet mass. Thus, the approximate equation is given by Bhaskara Rao et al. (1990) as,

$$\begin{aligned}\Delta g(x) = \gamma a_0\, \Delta x & \left[1n\, \frac{r-Y}{r+Y}\right]_{z=Z_1}^{Z_2} \\ & + 2\gamma a_1\, \Delta x \left[Y\, 1n\, 2(r+z) - x \arctan\frac{Yz}{xr}\right]_{z=Z_1}^{Z_2} \\ & + \gamma a_2\, \Delta x \left[2Yr - x^2 1n\, \frac{r-Y}{r+Y}\right]_{z=Z_1}^{Z_2},\end{aligned} \tag{3}$$

where $r = \sqrt{x^2 + Y^2 + z^2}$ and Δx is the station spacing. The corresponding equation in 2-D case



Fig. 2 - 2.5-D Prismatic model.

is given by

$$\begin{aligned}\Delta g(x) = {} & \gamma a_0 \ \Delta x \left[\ln r\right]_{z=Z_1}^{Z_2} \\ & + 2\gamma a_1 \ \Delta x \left[z - x \ \arctan \frac{z}{x} \right]_{z=Z_1}^{Z_2} \\ & + \gamma a_2 \ \Delta x \left[z^2 - x^2 \ \ln r \right]_{z=Z_1}^{Z_2}, \end{aligned} \qquad (4)$$

where $r = x^2 + z^2$ and Δx is the station spacing.

3. Application to 3-D modelling

Eqs. (2) and (3) are used to deduce the structure of a basin along a profile usually taken perpendicular to the elongation of the anomaly contours and passing through the minimum anomaly point, which may be called the principal profile. With a suitable modification of Eqs. (2) and (3), we can also deduce the structure of a basin along any other profile taken parallel to the principal profile. Thus, the equation for the gravity anomaly of a prismatic model, which is at an off-set distance of y from the principal profile, may be given by

$$\Delta g(x, y) = \frac{\Delta g(x, Y + y) + \Delta g(x, Y - y)}{2}. \qquad (5)$$



Fig. 3 - Structure derived by 2.5-D modelling of an off-set profile CD (Fig. 4) across the Los Angeles basin using a quadratic density function.

In other words, the gravity anomaly at any point is obtained by taking the average of the gravity anomalies calculated by replacing Y by $Y+y$ and $Y-y$, respectively, in Eqs. (2) and (3). The gravity anomaly along several profiles taken parallel to the principal profile at uniform off-set intervals covering the entire basin are interpreted by the above method, and the depths of the prisms along these profiles are deduced. The depths are used to construct a depth contour map, which may look like the structural map of a sedimentary basin. This method is comparable to the 3-D methods for deducing the structural map of a basin both in accuracy and computer time.

4. Computer programs

The computer program GR2HDSTR, coded in Fortran 77 for interpretation of gravity anomalies along the principal profile, is given in Appendix 1. This is used to calculate the basement topography of a sedimentary basin using a quadratic density function to account for variation of



Fig. 4 - Basement contour map of the Los Angeles basin derived from 2.5-D modelling of several off-set gravity profiles using a quadratic density function. Contour interval is 0.5 km.

density contrast with depth. A prism of finite strike length is assumed to be symmetrically placed below each observation point. The data is equispaced, and the width of the prisms is the same as the station interval. The average half strike length, *Y*, for each prism may be obtained by measuring the lengths of the elongation of the contours perpendicular to either side of the principal profile. *Y* may be different from one prism to another. The input consists of gravity anomalies, station spacing, number of observations, half strike length of each prism, and coefficients of quadratic density function. Total number of iterations and LT, the regions specifying the use of the exact and approximate equations, also may be specified as input.

Fig. 1 shows the linking of the subroutines, and the working of the algorithm is given below.

Main Program: The main program consists of input and output parameters and calling of various subroutines.

INIT: The subroutine subprogram 'INIT' calculates the initial depth to the bottom of the prism by using the Bouguer slab formula $Z(k)=\Delta g_{obs}(k) / 2\pi\gamma a_o$.

STRUC: The subroutine 'STRUC' calculates the gravity anomalies due to a given model of the sedimentary basin by appropriately using exact and approximate equations ANOM1 and ANOM2, respectively.

BOTT: The subroutine subprogram 'BOTT' iteratively improves the depths of the prisms by a method based on the differences between the observed and calculated anomalies. The increment or decrement to be given to the depth is

$$\Delta Z(k) = \left[\Delta g_{obs}(k) - \Delta g_{cal}(k)\right] / 2\pi\gamma\Delta\rho(z). \tag{6}$$



Fig. 5 - Structure derived by 2.5-D modelling of the principal gravity profile AB (Fig. 7) across the Pannonian basin using a quadratic density function. The corresponding interpretation with 2-D modelling is given by the dotted curve.

ANOM1: The function subprogram 'ANOM1' is used to calculate the gravity anomalies using the exact equation (Eq. (2)).

ANOM2: The function subprogram 'ANOM2' is used to calculate the gravity anomalies using the approximate equation (Eq. (3)).

RESUL: The subroutine 'RESUL' gives the results such as the distances from one end of the profile and the corresponding depths to the basement.



Fig. 6 - Structure derived by 2.5-D modelling of an off-set gravity profile (CD in Fig. 7) across the Pannonian basin using a quadratic density function.

The use of ANOM1 and ANOM2 in each stage is defined by the value of LT we assign. It has the value of LT1 (1) in the first stage, LT1 (2) in the second stage and LT1 (3) in the third stage. ANOM1 may be used to calculate the anomalies on the top and immediate neighbouring points of the prisms [LT1 (1)]. In the intermediate stage, the domain for the use of the exact equation may be extended to three points. In the last stage, the use of the exact equation may be extended to five stations from the centre of the prisms.



Fig. 7 - Basement contour map derived from 2.5-D modelling of several off-set profiles across the Pannonian basin using a quadratic density function. Contour interval is 0.2 km.

The programs work well for the situation of outcropping prisms by imposing a simple constraint Z_1=0.0001 km, to avoid any possible singularities. These programs can also be used without any modification for the situation of a constant density contrast or for a linear decrease of density contrast with depth by properly assigning zeros to a_1 and a_2.

The corresponding computer program GR2DSTR for 2-D modelling of sedimentary basins is given in Appendix 2. This program is supported by the subroutines INIT, STRUC, BOTT, ANOM1, ANOM2 and RESUL. The subroutines INIT, STRUC, BOTT and RESUL are the same as given in the program GR2HDSTR, and the listing of the subroutines ANOM1 and ANOM2 are given in Appendix 2.

A computer program GA3DSTR is also developed in Fortran 77 to deduce the structure of a sedimentary basin along any off-set profile and is given in Appendix 3. The only difference in this program from GR2HDSTR is that the values of half strike lengths are modified depending upon the value of the off-set distances, y, from the principal profile. Two half strike lengths are assigned to each prism by measuring the lengths of elongation of the contours on either side of the profile. These two values correspond to $Y+y$ and $Y-y$ in Eq. (5). The observed anomalies, Δg_{obs}, are measured along the off-set profile, and the calculated anomalies, Δg_{cal}, are obtained using Eq. (2) or Eq. (3), and the structure of the basin along the off-set profile is obtained using the differences between the observed and calculated anomalies as described above. By interpreting several off-set profiles covering the entire basin, we can deduce the 3D structural map of a basin. The program GA3DSTR is supported by a subroutine MEAN, along with the subroutines INIT, STRUC, BOTT and RESUL, and function subprograms ANOM1 and ANOM2 described in the program GR2HDSTR. The subroutine MEAN calculates the average gravity anomalies along any off-set profile using Eq. (5).

5. Examples of interpretation

The programs developed in this paper are applied to the interpretation of the residual gravity anomaly contour maps of the Los Angeles basin, California, (Chai and Hinze, 1988) and the Western part of the Pannonian basin, Eastern Austria (Granser, 1987). The density vs. depth data corresponding to the function $\Delta\rho\,(z)$=-0.5 exp (-0.1609 z) given by Chai and Hinze (1988) is fitted to a quadratic function by the least-squares method, and the coefficients are estimated (a_0=-0.49, a_1=0.067 and a_2 = -0.0028).

The residual gravity contour map published by Chai and Hinze (1988) has been interpreted by the 2.5-D method described in the text by taking the principal profile AB (Fig. 4) and several off-set profiles, such as CD (Fig. 4). The principal profile AB is sampled at equispaced intervals of 4 km, which is the width of each prism. The average half strike length of each prism is measured from the gravity contour map perpendicular to AB. The results of the inversion, and the corresponding error curve at the end of the 10th iteration are given by Bhaskara Rao et al. (1990). The depth section obtained by this method closely fits the section derived from the structural contour map published by Chai and Hinze (1988). The CPU time for 10 iterations on a PC 80386 computer using the program GR2HDSTR is about 1 min. However, if we carry out the inversion exclusively with the exact equation, the time would be about 2 min.

Fig. 3 shows the interpretation of a gravity anomaly profile taken along CD, which is parallel to the principal profile AB, and is at an off-set distance of 16 km, as shown in Fig. 4. This profile is interpreted using GA3DSTR and also by GR2DSTR. The depth section obtained along this profile by the present method and that taken from the structural map given by Chai and Hinze (1988) are also given in Fig. 3. These two interpretations are in close agreement.

The computer program GA3DSTR is also used to deduce the depths to the basement from the gravity anomalies along various off-set profiles taken at a uniform interval of 4 km, and covering the entire basin. Based on these interpreted depths, a depth contour map is drawn and is given in Fig. 4. This contour map is similar to the structural map derived by Chai and Hinze (1988). The advantage of analysing the gravity profiles with the present method is that different strike lengths of the prisms can be taken into account.

Fig. 5 represents another example of the inversion results for the principal profile AB (Fig. 7) in the Western part of the Pannonian basin, Eastern Austria (Granser, 1987). The density vs. depth data corresponding to the function $\Delta\rho$ (z)=-0.45 exp (-0.65 z) given by Granser (1987) is fitted to a quadratic function by the least-squares method, and the coefficients are estimated (a_0=-0.4386, a_1=0.2162 and a_2=-0.0297). The computer program GR2HDSTR is used to deduce the structure of the basin along this profile. The results obtained using the 2-D program GR2DSTR are also shown in Fig. 5, as are the depths to the basement taken from the structural map given by Granser (1987) along this profile. It may be observed that the depths obtained by the present methods agree with those given by Granser (1987).

Fig. 6 shows the interpretation of an off-set gravity anomaly profile taken along profile CD using the computer program GA3DSTR. The profile is at an off-set distance of 12 km from the principal profile as shown in Fig. 7. The anomalies are sampled at an interval of 2 km. Two half strike lengths are measured from the contour map on either side of the profile CD. The interpreted depths of the prisms are shown in Fig. 6. The depths obtained by Granser (1987) along this profile are also drawn by circles with continuous lines for comparision. These two results are in good agreement.

The computer program GA3DSTR is similarly used to deduce the depths to the basement from gravity anomalies along several profiles taken parallel to the principal profile at constant interval of 2 km, and covering the entire basin. The residuals are below ±0.1 mGal along any profile. The interpreted depths to the basement are drawn in the form of a depth contour map which is shown in Fig. 7. This map is similar to the structural map given by Granser (1987).

Acknowledgments. One of the authors (M. J. Prakash) is grateful to the Council of Scientific and Industrial Research, New Delhi, for providing financial assistance by way of Research Associate.

Appendix 1 - Computer program 6R2HDSTR coded in Fortran 77

```
******************************************************************************
      PROGRAM GR2HDSTR TO DEDUCE  THE  2.5-D  STRUCTURE  OF A
         SEDIMENTARY BASIN  FROM ITS GRAVITY  ANOMALIES  WITH  A
         QUADRATIC DENSITY FUNCTION.

         INPUT
         - - -

            GOBS     :  OBSERVED GRAVITY ANOMALIES (MGALS)
            NX       :  NUMBER OF OBSERVATIONS IN X-DIRECTION
            DX       :  STATION INTERVAL IN X-DIRECTION (KM)
            A0,A1&A2:  COEFFICIENTS OF QUADRATIC DENSITY
                        FUNCTION (GM/CC)
             ITER1   :  TOTAL NUMBER OF ITERATIONS REQUIRED
            ITR1        :   NUMBER OF ITERATIONS      REQUIRED IN FIRST  STAGE
            ITR2        :   NUMBER OF ITERATIONS REQUIRED IN SECOND STAGE
            LT1         :   LIMITING VALUE OF THE DOMAIN IN WHICH EXACT
                        EQUATION IS USED FOR ANOMALY CALCULATION
```

```
          Y        :  HALF STRIKE LENGTHS OF THE PRISMS (KM)

        OUTPUT
        - - - -

          GCAL       :   CALCULATED GRAVITY ANOMALIES (MGALS)
          Z          :   DEPTH TO THE BOTTOM (KM)

        SUPPORTING SUBROUTINES AND SUBPROGRAMS:
        - - - - - - - - - - - - - - - - - - - - - -
                           INIT, STRUC, BOTT, RESUL, ANOM1, ANOM2
******************************************************************************
******
        DIMENSION X(40),GOBS(40),GCAL(40),Z(40),Y(40),LT1(3)
        PI=3.14159265
        CONST=PI*40.0/3.0
           READ(*,801)NX,ITER1,DX,A0,A1,A2
        READ(*,802)ITR1,ITR2
        READ(*,803)(LT1(I),I=1,3)
        WRITE(*,901)NX,ITER1,DX,A0,A1,A2
        WRITE(*,902)ITR1,ITR2
        WRITE(*,903)(LT1(I),I=1,3)
        DO 10 I=1,NX
10      X(I)=FLOAT(I-1)*DX
        READ(*,804) (GOBS(I),I=1,NX)
        WRITE(*,904)(GOBS(I),I=1,NX)
        READ(*,804) (Y(I),I=1,NX)
        CALL INIT(NX,A0,X,GOBS,CONST,Z)
        WRITE(*,905) (Z(I),I=1,NX)
        WRITE(*,905) (Y(I),I=1,NX)
        DO 20 ITER = 1,ITER1
           CALL STRUC(NX,DX,A0,A1,A2,X,Y,Z,ITER,ITR1,ITR2,LT1,GCAL)
           CALL RESUL(NX,ITER,X,Z,GOBS,GCAL)
           CALL BOTT (NX,CONST,A0,A1,A2,GOBS,GCAL,Z)
20      CONTINUE
801     FORMAT(2I5,4F10.4)
802     FORMAT(2I5)
803     FORMAT(3I5)
804     FORMAT(8F10.4)
901     FORMAT(2I5,4F10.4)
902     FORMAT(2I5)
903     FORMAT(3I5)
904     FORMAT(8F10.4)
905     FORMAT(/8F10.4)
        STOP
        END
******************************************************************************
******
        SUBROUTINE INIT : CALCULATES THE INITIAL ESTIMATES OF
                          THE STRUCTURE.

        INPUT
        - - -
                      NX       :   NUMBER OF OBSERVATIONS IN X-DIRECTION
                      A0       :   CONSTANT DENSITY CONTRAST (GM/CC)
                      X        :   DISTANCE TO THE ANOMALY POINT IN X-DIRECTION
```

```
                    GOBS     :   OBSERVED GRAVITY ANOMALIES (MGALS)
                    CONST    :   PI*40.0/3.0

     OUTPUT
      - - - -
                    Z           :      DEPTH TO THE BOTTOM (KM)
******************************************************************************
******
        SUBROUTINE INIT(NX,A0,X,GOBS,CONST,Z)
        DIMENSION X(40),GOBS(40),Z(40)
        DO 10 I=1,NX
           Z(I)=GOBS(I)/(CONST*A0)
           IF (Z(I).LE.0.0)THEN
                Z(I)=0.00001
           ENDIF
10      CONTINUE
        RETURN
        END
******************************************************************************
******
        SUBROUTINE STRUC: CALCULATES THE  GRAVITY ANOMALIES  OF A
                          SEDIMENTARY  BASIN   BY  THE   COMBINED
                          USE OF  EXACT AND APPROXIMATE EQUATIONS
                          OF PRISMATIC MODEL.

     INPUT
     - - -
                  NX            :      NUMBER OF OBSERVATIONS IN X-DIRECTION
                  DX            :      STATION INTERVAL IN X-DIRECTION (KM)
                  A0,A1         :      COEFFICIENTS OF THE QUADRATIC DENSITY
                    & A2        FUNCTION (GM/CC)
                  X             :      DISTANCE  TO  THE  ANOMALY  POINT  IN  X-
DIRECTION
                  Z             :      DEPTH TO THE BOTTOM (KM)
                  ITER          :      ITERATION NUMBER
                  ITR1          :      NUMBER OF ITERATIONS REQUIRED IN FIRST
STAGE
                  ITR2          :      NUMBER OF ITERATIONS REQUIRED IN SECOND
STAGE
                  LT1           :      LIMITING VALUE OF THE DOMAIN IN WHICH
EXACT
                                    EQUATION IS USED FOR ANOMALY CALCULATION
                  Y             :      HALF STRIKE LENGTH OF THE PRISM (KM)

        OUTPUT
        - - - -
                  GCAL          :      CALCULATED GRAVITY ANOMALIES (MGALS)
******************************************************************************
******
        SUBROUTINE STRUC(NX,DX,A0,A1,A2,X,Y,Z,ITER,ITR1,ITR2,LT1,GCAL)
        DIMENSION X(40),Z(40),GCAL(40),Y(40),LT1(3)
        DO 10 I=1,NX
10      GCAL(I)=0.0
        Z1=0.0001
        IF (ITER.LT.ITR1)THEN
        LT=LT1(1)
```

```
        GOTO 50
        ENDIF
        IF (ITER.LE.ITR2)THEN
        LT=LT1(2)
        GOTO 50
        ELSE
        LT=LT1(3)
        ENDIF
50      DO 20 I=1,NX
          DO 20 K=1,NX
                X1=X(I)-X(K)
                Z2=Z(K)
                Y1=Y(K)
                IF (ABS(I-K).LE.LT)THEN
                    GCAL1=ANOM1(DX,Y1,X1,Z1,Z2,A0,A1,A2)
                    GOTO 100
                ELSE
                    GCAL1=ANOM2(DX,Y1,X1,Z1,Z2,A0,A1,A2)
                ENDIF
100                  GCAL(I)=GCAL(I)+GCAL1
20      CONTINUE
        RETURN
        END
*******************************************************************************
******
        SUBROUTINE BOTT: CALCULATES THE STRUCTURE OF A
                         BASIN  USING THE
                         BOTT'S METHOD.

        INPUT
      - - -
                      NX           :      NUMBER OF OBSERVATIONS IN X-DIRECTION
                      A0,A1        :      COEFFICIENTS OF QUADRATIC DENSITY
                        & A2              FUNCTION (GM/CC)
                      GOBS         :      OBSERVED   GRAVITY ANOMALIES (MGALS)
                      GCAL         :      CALCULATED GRAVITY ANOMALIES (MGALS)
                      CONST        :      PI*40.0/3.0

        OUTPUT
      - - - -
                      Z            :      DEPTH TO THE BASEMENT (KM)
*******************************************************************************
******
        SUBROUTINE BOTT(NX,CONST,A0,A1,A2,GOBS,GCAL,Z)
        DIMENSION GOBS(40),GCAL(40),Z(40),DZ(40)
        DO 10 I=1,NX
           R2=A0+A1*Z(I)+A2*Z(I)*Z(I)
        CONST1=CONST*R2
10      DZ(I)=(GOBS(I)-GCAL(I))/CONST1
        DO 20 I=1,NX
           Z(I)=Z(I)+DZ(I)
           IF (Z(I).LE.0.0)THEN
                        Z(I)=0.00001
           ENDIF
20      CONTINUE
        RETURN
```

```
        END
*******************************************************************************
******
      FUNCTION SUBPROGRAM ANOM1: CALCULATES THE ANOMALY  OF
                                 PRISMATIC MODEL WITH EXACT
                                @EQUATION.

      INPUT
      - - -
                   DX       :      STATION INTERVAL IN X-DIRECTION (KM)
                   A0,A1    :      COEFFICIENTS OF QUADRATIC DENSITY
                     & A2                 FUNCTION (GM/CC)
                   X1       :      DISTANCE TO THE POINT OF CALCULATION
                                          IN X-DIRECTION WITH REFERENCE TO THE
                                          EPICENTRE OF THE PRISM UNDER CONSIDE-
RATION
                   Y1       :      HALF STRIKE LENGTH OF THE PRISM  (KM)
                   Z1       :      DEPTH TO THE TOP    OF THE PRISM (KM)
                   Z2       :      DEPTH TO THE BOTTOM OF THE PRISM (KM)
                   T        :      HALF THICKNESS OF THE PRISM (KM)
                   GAMA     :      GRAVITATIONAL CONSTANT
*******************************************************************************
******
        FUNCTION ANOM1(DX,Y1,X1,Z1,Z2,A0,A1,A2)
        T=DX/2.0
        GAMA=20.0/3.0
        F1=X1-T
        F2=X1+T
           IF(F1.EQ.0.0)THEN
               F1=0.00001
        ENDIF
        IF(F2.EQ.0.0)THEN
           F2=0.00001
        ENDIF
        IF(Z1.EQ.0.0)THEN
            Z1=0.00001
        ENDIF
        R1=SQRT(Z1*Z1+F1*F1+Y1*Y1)
        R2=SQRT(Z2*Z2+F1*F1+Y1*Y1)
        R3=SQRT(Z1*Z1+F2*F2+Y1*Y1)
        R4=SQRT(Z2*Z2+F2*F2+Y1*Y1)
        T1=ATAN((Y1*F1)/(Z1*R1))-ATAN((Y1*F2)/(Z1*R3))
        T2=ATAN((Y1*F2)/(Z2*R4))-ATAN((Y1*F1)/(Z2*R2))
        T3=ALOG(((R2+F1)*(R3+F2))/((R1+F1)*(R4+F2)))
        T4=ALOG(((R2+Y1)*SQRT(Z1*Z1+F1*F1))/((R1+Y1)*SQRT(Z2*Z2+F1*F1)))
        T5=ALOG(((R3+Y1)*SQRT(Z2*Z2+F2*F2))/((R4+Y1)*SQRT(Z1*Z1+F2*F2)))
        T6=ALOG((R1+Z1)/(R2+Z2))
        T7=ATAN((F1*Z2)/(Y1*R2))-ATAN((F1*Z1)/(Y1*R1))
        T8=ATAN((F1*R1)/(Y1*Z1))-ATAN((F1*R2)/(Y1*Z2))
        T9=ALOG((R4+Z2)/(R3+Z1))
        T10=ATAN((F2*Z2)/(Y1*R4))-ATAN((F2*Z1)/(Y1*R3))
        T11=ATAN((F2*R3)/(Y1*Z1))-ATAN((F2*R4)/(Y1*Z2))
        P1=2*GAMA*(Z1*T1+Z2*T2+Y1*T3+F1*T4+F2*T5)
        P2=GAMA*((Z1*Z1)*T1+(Z2*Z2)*T2)
        P31=2*Y1*F1*T6+Y1*Y1*T7+F1*F1*T8
        P3=GAMA*P31
```

```
        P41=2*Y1*F2*T9-Y1*Y1*T10-F2*F2*T11
        P4=GAMA*P41
        P51=((Z1**3)*T1/3.0+(Z2**3)*T2/3.0)
        P5=2*GAMA*P51
        P61=(2*Y1*(R4-R3)/3.0-(F2*F2)*T5/3.0)
        P6=2*GAMA*F2*P61
        P71=(2*Y1*(R2-R1)/3.0+(F1*F1)*T4/3.0)
        P7=2*GAMA*F1*P71
        P8=2*GAMA*Y1**3*(T3/3.0)
        TA0=P1
        TA1=P2+P3+P4
        TA2=P5+P6-P7-P8
        ANOM1=A0*TA0+A1*TA1+A2*TA2
        RETURN
        END
*****************************************************************************
******
        FUNCTION SUBPROGRAM ANOM2: CALCULATES THE ANOMALY OF
                                  PRISMATIC   MODEL    WITH
                                  APPMATE EQUATION.

        INPUT
        - - -
                   DX      :      STATION INTERVAL IN X-DIRECTION (KM)
                   A0,A1   :      COEFFICIENTS OF QUADRATIC DENSITY
                   & A2                 FUNCTION (GM/CC)
                   X1      :      DISTANCE TO THE POINT OF CALCULATION
                                        IN X-DIRECTION WITH REFERENCE TO THE
                                        EPICENTRE OF THE PRISM UNDER CONSIDE-
RATION
                   Z1      :      DEPTH TO THE TOP    OF THE PRISM (KM)
                   Z2      :      DEPTH TO THE BOTTOM OF THE PRISM (KM)
                   Y1      :      HALF STRIKE LENGTH  OF THE PRISM (KM)
                   GAMA    :      GRAVITATIONAL CONSTANT
*****************************************************************************
******
        FUNCTION ANOM2(DX,Y1,X1,Z1,Z2,A0,A1,A2)
        GAMA=20.0/3.0
        CONST=2.0*GAMA*Y1*DX
        IF (X1.EQ.0.0)THEN
                    X1=0.1
        ENDIF
        R1=X1*X1+Y1*Y1+Z1*Z1
        R1=SQRT(R1)
        P5=R1-Y1
        IF (ABS(P5).LE.0.1)THEN
                P5=0.15
        ENDIF
        R2=X1*X1+Y1*Y1+Z2*Z2
        R2=SQRT(R2)
        P1=ALOG((R2-Y1)*(R1+Y1)/((R2+Y1)*P5))
        P2=ALOG((R2+Z2)/(R1+Z1))
        P3=ATAN(Y1*Z2/(X1*R2))-ATAN(Y1*Z1/(X1*R1))
        TP1=CONST*A0*P1/(2.0*Y1)
        TP2=CONST*A1*(P2-(X1/Y1)*P3)
        TP3=CONST*A2*(R2-R1-X1*X1*P1/(2.0*Y1))
```

```
        ANOM2=TP1+TP2+TP3
           RETURN
        END
********************************************************************************
******
        SUBROUTINE RESUL: GIVES  THE   RESULTS  SUCH  AS  DEPTHS
                         TO THE BASEMENT AND RESIDUAL ANOMALIES.

        INPUT
        - - -
                  NX             :      NUMBER OF OBSERVATIONS IN X-DIRECTION
                  ITER           :      ITERATION NUMBER
                  X              :        DISTANCE TO THE ANOMALY POINT IN X-
DIRECTION
                  Z              :      DEPTH TO THE BASEMENT (KM)
                  GOBS           :      OBSERVED GRAVITY ANOMALIES (MGALS)

        OUTPUT
        - - - -
                    GCAL         :      CALCULATED GRAVITY ANOMALIES (MGALS)
********************************************************************************
******
        SUBROUTINE RESUL(NX,ITER,X,Z,GOBS,GCAL)
        DIMENSION X(40),GOBS(40),GCAL(40),RESD(40),Z(40)
        DO 10 I=1,NX
10      RESD(I)=GOBS(I)-GCAL(I)
        WRITE(*,906) ITER
        WRITE(*,907)
        DO 20 I=1,NX
           WRITE(*,908) X(I),Z(I),GOBS(I),GCAL(I),RESD(I)
20      CONTINUE
906     FORMAT(/5X,'ITERATION NO.=',I4)
907     FORMAT(/7X,'DISTANCE',9X,'DEPTH',8X,'OBS ANOM',7X,'CAL ANOM'
           *,10X,'RESD',//)
908        FORMAT(5(5X,F10.4))
        RETURN
        END
```

Appendix 2 - Computer program GR2DSTR coded in Fortran 77

```
********************************************************************************
******
        PROGRAM GR2DSTR TO DEDUCE  THE  2-D  STRUCTURE  OF A
        SEDIMENTARY BASIN  FROM ITS GRAVITY  ANOMALIES  WITH
        A QUADRATIC DENSITY FUNCTION.

        INPUT
        - - -

           GOBS       :   OBSERVED GRAVITY ANOMALIES (MGALS)
           NX         :   NUMBER OF OBSERVATIONS IN X-DIRECTION
           DX         :   STATION INTERVAL IN X-DIRECTION (KM)
        A0,A1&A2   :  COEFFICIENTS OF QUADRATIC DENSITY
```

```
                  FUNCTION (GM/CC)
      ITER1     : TOTAL NUMBER OF ITERATIONS REQUIRED
      ITR1      : NUMBER OF ITERATIONS REQUIRED IN FIRST  STAGE
      ITR2      : NUMBER OF ITERATIONS REQUIRED IN SECOND STAGE
      LT1       : LIMITING VALUE OF THE DOMAIN IN WHICH EXACT
                   EQUATION IS USED FOR ANOMALY CALCULATION

      OUTPUT
      - - - -

      GCAL      : CALCULATED GRAVITY ANOMALIES (MGALS)
      Z         : DEPTH TO THE BOTTOM (KM)

      SUPPORTING SUBROUTINES AND SUBPROGRAMS:
      - - - - - - - - - - - - - - - - - - - - - - - -
                               INIT, STRUC, BOTT, RESUL, ANOM1, ANOM2
******************************************************************************
******
      DIMENSION X(40),GOBS(40),GCAL(40),Z(40),LT1(3)
      PI=3.14159265
      CONST=PI*40.0/3.0
      READ(*,801)NX,ITER1,DX,A0,A1,A2
      READ(*,802)ITR1,ITR2
      READ(*,803)(LT1(I),I=1,3)
      WRITE(*,901)NX,ITER1,DX,A0,A1,A2
      WRITE(*,902)ITR1,ITR2
      WRITE(*,903)(LT1(I),I=1,3)
      DO 10 I=1,NX
10    X(I)=FLOAT(I-1)*DX
      READ(*,804) (GOBS(I),I=1,NX)
      WRITE(*,904)(GOBS(I),I=1,NX)
      CALL INIT(NX,A0,X,GOBS,CONST,Z)
      WRITE(*,905) (Z(I),I=1,NX)
      DO 20 ITER = 1,ITER1
        CALL STRUC(ITER,ITR1,ITR2,LT1,NX,DX,A0,A1,A2,X,Z,GCAL)
        CALL RESUL(NX,ITER,X,Z,GOBS,GCAL)
        CALL BOTT (NX,CONST,A0,A1,A2,GOBS,GCAL,Z)
20    CONTINUE
801   FORMAT(2I5,4F10.4)
802   FORMAT(2I5)
803   FORMAT(3I5)
804   FORMAT(8F10.4)
901   FORMAT(2I5,4F10.4)
902   FORMAT(2I5)
903   FORMAT(3I5)
904   FORMAT(8F10.4)
905   FORMAT(8F10.4)
      STOP
      END
******************************************************************************
******
      FUNCTION SUBPROGRAM ANOM1: CALCULATES THE ANOMALYOF
                              2-D PRISMATIC MODEL WITH EXACT EQUATION.

      INPUT
    - - -
```

```
                DX      :      STATION INTERVAL IN X-DIRECTION (KM)
                A0,A1   :      COEFFICIENTS OF QUADRATIC DENSITY
                & A2                  FUNCTION (GM/CC)
                X1      :      DISTANCE TO THE POINT OF CALCULATION
                                      IN X-DIRECTION WITH REFERENCE TO THE
                                      EPICENTRE OF THE PRISM UNDER CONSIDE-
RATION
                Z1      :      DEPTH TO THE TOP    OF THE PRISM (KM)
                Z2      :      DEPTH TO THE BOTTOM OF THE PRISM (KM)
                T       :      HALF THICKNESS OF THE PRISM (KM)
                GAMA    :      GRAVITATIONAL CONSTANT
*****************************************************************************
******
        FUNCTION ANOM1(DX,X1,Z1,Z2,A0,A1,A2)
        PI=3.14159265
        T=DX/2.0
        GAMA=20.0/3.0
        CONST=2*GAMA
        F1=X1-T
        F2=X1+T
        IF(F1 .EQ. 0.0) THEN
          F1=0.0001
        ENDIF
        IF(F2 .EQ. 0.0) THEN
               F2=0.0001
        ENDIF
        IF(Z1 .EQ. 0.0) THEN
          Z1=0.0001
        ENDIF
        R1=SQRT(Z1*Z1+F2*F2)
        R2=SQRT(Z1*Z1+F1*F1)
        R3=SQRT(Z2*Z2+F1*F1)
        R4=SQRT(Z2*Z2+F2*F2)
        R41=ALOG(R4/R1)
        R32=ALOG(R3/R2)
        PHI1=PI/2+ATAN(F2/Z1)
        PHI2=PI/2+ATAN(F1/Z1)
        PHI3=PI/2+ATAN(F1/Z2)
        PHI4=PI/2+ATAN(F2/Z2)
        PHI43=PHI4-PHI3
        PHI12=PHI1-PHI2
        PHI23=PHI2-PHI3
        PHI14=PHI1-PHI4
        TP1=Z2*PHI43-Z1*PHI12+F2*R41-F1*R32
        TP1=TP1*CONST*A0
        TP2=Z2*Z2*PHI43-Z1*Z1*PHI12+F1*F1*PHI23-F2*F2*PHI14+2*T*(Z2-Z1)
        TP2=TP2*CONST*A1/2.0
        TP3=Z2**3*PHI43-Z1**3*PHI12+F1**3*R32-F2**3*R41+T*(Z2*Z2-Z1*Z1)
        TP3=TP3*CONST*A2/3.0
        ANOM1=TP1+TP2+TP3
        RETURN
        END
*****************************************************************************
******    FUNCTION SUBPROGRAM ANOM2: CALCULATES THE ANOMALY OF
                             2-D PRISMATIC  MODEL WITH
```

```
                              APPROXIMATE EQUATION.

*         INPUT                                                        *
          - - -
                DX       :       STATION INTERVAL IN X-DIRECTION (KM)
                A0,A1    :       COEFFICIENTS OF QUADRATIC DENSITY
                  & A2           FUNCTION (GM/CC)
                X1       :       DISTANCE TO THE POINT OF CALCULATION
                                        IN X-DIRECTION WITH REFERENCE TO THE
                                        EPICENTRE OF THE PRISM UNDER CONSIDE-
RATION
                Z1       :       DEPTH TO THE TOP    OF THE PRISM (KM)
                Z2       :       DEPTH TO THE BOTTOM OF THE PRISM (KM)
                GAMA     :       GRAVITATIONAL CONSTANT
*****************************************************************************
******
        FUNCTION ANOM2(DX,X1,Z1,Z2,A0,A1,A2)
        GAMA=20.0/3.0
        CONST=2.0*GAMA*DX
        IF (X1 .EQ. 0.0) THEN
                  X1=0.0001
        ENDIF
        R1=X1*X1+Z1*Z1
        R1=SQRT(R1)
        R2=X1*X1+Z2*Z2
        R2=SQRT(R2)
        R21=ALOG(R2/R1)
        PHI21=ATAN(Z2/X1)-ATAN(Z1/X1)
        TP1=CONST*A0*R21
        TP2=CONST*A1*((Z2-Z1)-X1*PHI21)
        TP3=CONST*A2*((Z2*Z2-Z1*Z1)*0.5-X1*X1*R21)
        ANOM2=TP1+TP2+TP3
        RETURN
           END
```

Appendix 3 - Computer program GA3DSTR coded in Fortran 77

```
*****************************************************************************
******          PROGRAM  GA3DSTR   TO  DEDUCE THE  2.5-D  STRUCTURE OF A
        SEDIMENTARY BASIN  ALONG ANY  OFF-SET  PROFILE FROM  ITS
        GRAVITY   ANOMALIES  WITH A  QUADRATIC DENSITY FUNCTION.

        INPUT
        - - -

          GOBS          :       OBSERVED GRAVITY ANOMALIES (MGALS)
          NX            :       NUMBER OF OBSERVATIONS IN X-DIRECTION
          DX            :       STATION INTERVAL IN X-DIRECTION (KM)
          DY            :       SCALE CONSTANT IN Y-DIRECTION   (KM)
          A0,A1&A2   :   COEFFICIENTS OF QUADRATIC DENSITY
                         FUNCTION (GM/CC)
          ITER1      :   TOTAL NUMBER OF ITERATIONS REQUIRED
```

```
          ITR1            :       NUMBER OF ITERATIONS REQUIRED IN FIRST  STAGE
          ITR2            : NUMBER OF ITERATIONS REQUIRED IN SECOND STAGE
          LT1             :       LIMITING VALUE OF THE DOMAIN IN WHICH EXACT
                          EQUATION IS USED FOR ANOMALY CALCULATION
          YY              :       STRIKE LENGTHS ON EITHER SIDE OF THE PROFILE
                          CORRESPONDING TO Y+y AND Y-y

        OUTPUT
        - - - -

          GCAL            :       CALCULATED GRAVITY ANOMALIES (MGALS)
          Z           :   DEPTH TO THE BOTTOM (KM)

        SUPPORTING SUBROUTINES AND SUBPROGRAMS:
        - - - - - - - - - - - - - - - - - - - - - - - -
                          MEAN,INIT, STRUC, BOTT, RESUL, ANOM1, ANOM2
******************************************************************************
******               DIMENSION X(40),GOBS(40),GCAL(40),Z(40),YY(2,40),LT1(3)
        PI=3.14159265
        CONST=PI*40.0/3.0
        READ(*,801)NX,ITER1,DX,DY,A0,A1,A2
        READ(*,802)ITR1,ITR2
        READ(*,803)(LT1(I),I=1,3)
        WRITE(*,901)NX,ITER1,DX,DY,A0,A1,A2
        WRITE(*,902)ITR1,ITR2
        WRITE(*,903)(LT1(I),I=1,3)
        DO 10 I=1,NX
10      X(I)=FLOAT(I-1)*DX
        READ(*,804) (GOBS(I),I=1,NX)
        WRITE(*,904)(GOBS(I),I=1,NX)
        DO 20 I=1,2
        READ(*,804) (YY(I,J),J=1,NX)
20      CONTINUE
        DO 30 I=1,2
           DO 30 J=1,NX
        YY(I,J)=YY(I,J)*DY
30      CONTINUE
        CALL INIT(NX,A0,X,GOBS,CONST,Z)
        WRITE(*,905) (Z(I),I=1,NX)
        DO 40 I=1,2
        WRITE(*,905) (YY(I,J),J=1,NX)
40      CONTINUE
        DO 50 ITER = 1,ITER1
           CALL MEAN(NX,DX,A0,A1,A2,X,YY,Z,ITER,ITR1,ITR2,LT1,GCAL)
           CALL RESUL(NX,ITER,X,Z,GOBS,GCAL)
           CALL BOTT (NX,CONST,A0,A1,A2,GOBS,GCAL,Z)
50      CONTINUE
801     FORMAT(2I5,5F10.4)
802     FORMAT(2I5)
803     FORMAT(3I5)
804     FORMAT(8F10.4)
901     FORMAT(2I5,5F10.4)
902     FORMAT(2I5)
903     FORMAT(3I5)
904     FORMAT(8F10.4)
905     FORMAT(/8F10.4)
```

```
        STOP
        END
*******************************************************************************
******
        SUBROUTINE  MEAN: CALCULATES THE  MEAN GRAVITY ANOMALY
                        USING STRIKE LENGTHS OF Y+y AND Y-y

        INPUT
        - - -
                  GOBS           :      OBSERVED GRAVITY ANOMALIES (MGALS)
                  NX             :      NUMBER OF OBSERVATIONS IN X-DIRECTION
                  DX             :      STATION INTERVAL IN X-DIRECTION (KM)
                  A0,A1          :      COEFFICIENTS OF THE QUADRATIC DENSITY
                  & A2          FUNCTION (GM/CC)
                  ITER           :      ITERATION NUMBER
                  ITR1           :      NUMBER OF ITERATIONS REQUIRED IN FIRST
STAGE
                  ITR2           :      NUMBER OF ITERATIONS REQUIRED IN SECOND
STAGE
                  LT1     :     LIMITING VALUE OF THE DOMAIN IN WHICH EXACT
                                  EQUATION IS USED FOR ANOMALY CALCULATION
                  X              :      DISTANCE  TO  THE  ANOMALY  POINT  IN  X-
DIRECTION
                  YY             :      STRIKE  LENGTHS  ON  EITHER  SIDE  OF  THE
PROFILE
                                       CORRESPONDING TO Y+y AND Y-y
                  Z              :      DEPTH TO THE BOTTOM (KM)

        OUTPUT
        - - - -
                     GCAL        :      CALCULATED GRAVITY ANOMALIES (MGALS)
*******************************************************************************
******
        SUBROUTINE MEAN (NX,DX,A0,A1,A2,X,YY,Z,ITER,ITR1,ITR2,LT1,GCAL)
        DIMENSION X(40),Z(40),Y(40),YY(2,40),GCAL(40),GG(2,40),G1(40)
        DO 10 I=1,2
          DO 20 J=1,NX
20         Y(J)=YY(I,J)
          CALL STRUC (NX,DX,A0,A1,A2,X,Y,Z,ITER,ITR1,ITR2,LT1,G1)
               DO 30 K=1,NX
30             GG(I,K)=G1(K)
10      CONTINUE
        DO 40 I=1,NX
          GCAL (I) =(GG(1,I)+GG(2,I))/2.0
40      CONTINUE
        RETURN
        END
```

References

Bhaskara Rao D.; 1986: *Modelling of sedimentary basins from gravity anomalies with variable density contrast.* Geophy. Jour. Roy. Astr. Soc., **84**, 207-212.

Bhaskara Rao D., Prakash M. J. and Ramesh Babu N.; 1990: *3D and 2-1/2-D modelling of gravity anomalies with* *variable density contrast.* Geophys. Prosp., **38**, 411-422.

Bott M. H. P.; 1960: *The use of rapid digital computing methods for direct gravity interpretation of sedimentary basins.* Geophy. Jour. Roy. Astr. Soc., **3**, 63-67.

Chai Y. and Hinze W. J.; 1988: *Gravity inversion of an interface above which the density contrast varies exponentially with depth.* Geophysics, **53**, 837-845.

Cordell L. and Henderson R. G.; 1968: *Iterative three-dimensional solution of gravity anomaly data using a digital computer.* Geophysics, **33**, 596-601.

Enmark T.; 1981: *A versatile interactive computer program for computation and automatic optimization of gravity models.* Geoexploration, **19**, 47-66.

Goetze H. J. and Lahmeyer B.; 1988: *Application of three-dimensional interactive modeling in gravity and magnetics.* Geophysics, **53**, 1096-1108.

Granser H.; 1987: *Three-dimensional interpretation of gravity data from sedimentary basins using an exponential density-depth function.* Geophys. Prosp., **35**, 1030-1041.

Nagy D.; 1966: *The gravitational attraction of a right rectangular prism.* Geophysics, **31**, 362-371.

Rasmussen R. and Pedersen L. B.; 1979: *End corrections in potential field modelling.* Geophys. Prosp. **27**, 749-760.

Talwani M., Worzel J. L. and Landisman M.; 1959: *Rapid gravity computations for two-dimensional bodies with application to the Meandocino Submarine Fracture zone.* J. Gophys. Res., **64**, 49-59.

Talwani M. and Ewing M.; 1960: *Rapid computation of gravitational attraction of three-dimensional bodies of arbitrary shape.* Geophysics, **25**, 203-225.